General Relativistic Calculations in *Mathematica*

By George E. Hrabovsky

What I Will Cover

- **What is xAct?**
- **What Do We Do Next?**
- **Contraction**
- **Covariant Derivatives**
- **The Second Bianchi Identity**
- **Perturbative GR**

What is xAct?

This is a very complete system for doing tensor analysis in the *Mathematica* system.

All you need do is open the system:

```
<< xAct`xTensor`
```

```
------------------------------------------------------------

Package xAct`xPerm`  version 1.2.2, {2014, 9, 28}

CopyRight (C) 2003-2014, Jose M. Martin-Garcia, under the General Public License.

Connecting to external MinGW executable...

Connection established.

------------------------------------------------------------

Package xAct`xTensor`  version 1.1.1, {2014, 9, 28}

CopyRight (C) 2002-2014, Jose M. Martin-Garcia, under the General Public License.

------------------------------------------------------------

These packages come with ABSOLUTELY NO WARRANTY; for details type
  Disclaimer[]. This is free software, and you are welcome to redistribute
  it under certain conditions. See the General Public License for details.

------------------------------------------------------------
```

What Do We Do Next?

Now that we have started things up, there are some things we need to do.

1. **We need to define the underlying manifold we are working in. We do this with the DefManifold command: DefManifold[M, dim, {a, b, c,...}] defines M to be an n-dimensional differentiable manifold with dimension dim (a positive integer or a constant symbol) and tensor abstract indices a, b, c, ... DefManifold[M, {M1, ..., Mm}, {a, b, c,...}] defines M to be the product manifold of previously defined manifolds M1 ... Mm. For backward compatibility dim can be a list of positive integers, whose length is interpreted as the dimension of the defined manifold. Here we define a four-dimensional manifold having the indices $a, b, c, d, e, f, g, h, i, j, k, l$:**

```
DefManifold[M, 4, {a, b, c, d, e, f, g, h, i, j, k, l}]
** DefManifold: Defining manifold M.
** DefVBundle: Defining vbundle TangentM.
```

In defining the manifold, we have also defined the *vector bundle* called TangentM, $\pi : TM \to M$. This is where our tensors will be.

2. **We need to define the metric using the DefMetric command: DefMetric[signdet, metric[-a,-b], covd, covdsymbol] defines metric[-a, -b] with signdet 1 or -1 and associates the covariant derivative covd[-a] to it. Note that we have the convention [-a] has a as the subscript, and [a] has it as the superscript.**

```
DefMetric[-1, metric[-i, -j], cd, {";", "∇"}, PrintAs → "g"]
```

```
** DefTensor: Defining symmetric metric tensor metric[-i, -j].
** DefTensor: Defining antisymmetric tensor epsilonmetric[-a, -b, -c, -d].
** DefTensor: Defining tetrametric Tetrametric[-a, -b, -c, -d].
** DefTensor: Defining tetrametric Tetrametric†[-a, -b, -c, -d].
** DefCovD: Defining covariant derivative cd[-i].
** DefTensor: Defining vanishing torsion tensor Torsioncd[a, -b, -c].
** DefTensor: Defining symmetric Christoffel tensor Christoffelcd[a, -b, -c].
** DefTensor: Defining Riemann tensor Riemanncd[-a, -b, -c, -d].
** DefTensor: Defining symmetric Ricci tensor Riccicd[-a, -b].
** DefCovD:  Contractions of Riemann automatically replaced by Ricci.
** DefTensor: Defining Ricci scalar RicciScalarcd[].
** DefCovD:  Contractions of Ricci automatically replaced by RicciScalar.
** DefTensor: Defining symmetric Einstein tensor Einsteincd[-a, -b].
** DefTensor: Defining Weyl tensor Weylcd[-a, -b, -c, -d].
** DefTensor: Defining symmetric TFRicci tensor TFRiccicd[-a, -b].
** DefTensor: Defining Kretschmann scalar Kretschmanncd[].
** DefCovD:  Computing RiemannToWeylRules for dim 4
** DefCovD:  Computing RicciToTFRicci for dim 4
** DefCovD:  Computing RicciToEinsteinRules for dim 4
** DefTensor: Defining weight +2 density Detmetric[]. Determinant.
```

3. **We can define tensors using the DefTensor command: DefTensor[T[-a, b, c, ...], {M1, ...}] defines T to be a tensor field on manifolds and parameters M1,... and the base manifolds associated to the vector bundles of its indices. DefTensor[T[-a, b, c, ...], {M1, ...}, symmetry] defines a tensor with symmetry given by a generating set or strong generating set of the associated permutation group. In fact we can define any order of tensor just by specifying the indices. A scalar has no indices so we can define the scalar**

```
DefTensor[s[], M]
```

```
** DefTensor: Defining tensor s[].
```

```
s
```

```
s
```

a tangent vector

```
DefTensor[tv[i], M]
```

```
** DefTensor: Defining tensor tv[i].
```

```
tv[i]
```

tv^{i}

a covector

```
DefTensor[cv[-i], M]
```

```
** DefTensor: Defining tensor cv[-i].
```

```
cv[-i]
```

cv_i

and so on.

We can define tensors by their symmetries or antisymmetries, too. Here we define the tensor T_{ij} that is antisymmetric:

```
DefTensor[T[-i, -j], M, Antisymmetric[{-i, -j}]];
```

```
** DefTensor: Defining tensor T[-i, -j].
```

```
T[-i, -j]
```

T_{ij}

```
T[-j, -i]
```

T_{ji}

To make this work, since *Mathematica* does not enforce our symmetry rules, we need to make it do so by using the ToCanonical command. Let's try it again

```
T[-j, -i] // ToCanonical
```

$-T_{ij}$

We can add two such tensors in a way that gives us a 0 result,

```
T[-i, -j] + T[-j, -i]
```

$T_{ij} + T_{ji}$

```
T[-i, -j] + T[-j, -i] // ToCanonical
```

0

Contraction

We can do some operations. Let’s say we have the Riemann tensor R_{ijkl} and we want to contract it using the metric g^{ik}. we use the ContractMetric command,

```
metric[i, k] Riemanncd[-i, -j, -k, -l] // ContractMetric
```

```
R[∇]_jl
```

Mathematica also understands that this is the Ricci tensor, R_{jl},

```
metric[i, k] Riemanncd[-i, -j, -k, -l] // ContractMetric // InputForm
```

```
Riccicd[-j, -l]
```

Contracting it again gives us the Ricci scalar,

```
metric[j, l] Riccicd[-j, -l] // ContractMetric // InputForm
```

```
RicciScalarcd[]
```

Covariant Derivatives

The covariant derivative of our tensor, $\nabla_i T_{jl}$, is input

```
cd[-i][T[-j, -l]]
```

$\nabla_i T_{jl}$

If we have multiple covariant derivatives, we would enter them as follows, where @ is the Map command:

```
cd[-a]@cd[-b]@cd[-c]@T[-d, -e] // ToCanonical
```

$\nabla_a \nabla_b \nabla_c T_{de}$

To evaluate this we use the command SortCovDs

```
cd[-a]@cd[-b]@cd[-c]@T[-d, -e] // ToCanonical // SortCovDs
```

$$- R[\nabla]_{cbe}{}^{l\$1032} \nabla_a T_{dl\$1032} - R[\nabla]_{cbd}{}^{l\$1032} \nabla_a T_{l\$1032e} - T_{l\$1030e} \nabla_b R[\nabla]_{cad}{}^{l\$1030} - T_{dl\$1030} \nabla_b R[\nabla]_{cae}{}^{l\$1030} - R[\nabla]_{cae}{}^{l\$1030} \nabla_b T_{dl\$1030} - R[\nabla]_{cad}{}^{l\$1030} \nabla_b T_{l\$1030e} - R[\nabla]_{bae}{}^{l\$1028} \nabla_c T_{dl\$1028} - R[\nabla]_{bad}{}^{l\$1028} \nabla_c T_{l\$1028e} + \nabla_c \nabla_b \nabla_a T_{de} - R[\nabla]_{bac}{}^{l\$1028} \nabla_{l\$1028} T_{de} - R[\nabla]_{cba}{}^{l\$1032} \nabla_{l\$1032} T_{de}$$

This is correct, but too difficult to interpret, we add the command ScreenDollarIndices to put it into a lnaguage we can read instead of just the computer reading it

```
cd[-a]@cd[-b]@cd[-c]@T[-d, -e] // ToCanonical // SortCovDs // ScreenDollarIndices
```

$-R[\nabla]_{cbe}{}^{f}\,\nabla_a T_{df} - R[\nabla]_{cbd}{}^{f}\,\nabla_a T_{fe} - T_{fe}\,\nabla_b R[\nabla]_{cad}{}^{f} - T_{df}\,\nabla_b R[\nabla]_{cae}{}^{f} - R[\nabla]_{cae}{}^{f}\,\nabla_b T_{df} - R[\nabla]_{cad}{}^{f}\,\nabla_b T_{fe} - R[\nabla]_{bae}{}^{f}\,\nabla_c T_{df} - R[\nabla]_{bad}{}^{f}\,\nabla_c T_{fe} + \nabla_c\nabla_b\nabla_a T_{de} - R[\nabla]_{bac}{}^{f}\,\nabla_f T_{de} - R[\nabla]_{cba}{}^{f}\,\nabla_f T_{de}$

Instead of having to write all of this all of the time we make a session-wide command

```
$PrePrint = ScreenDollarIndices;
```

```
cd[-a]@cd[-b]@cd[-c]@T[-d, -e] // ToCanonical // SortCovDs
```

$-R[\nabla]_{cbe}{}^{f}\,\nabla_a T_{df} - R[\nabla]_{cbd}{}^{f}\,\nabla_a T_{fe} - T_{fe}\,\nabla_b R[\nabla]_{cad}{}^{f} - T_{df}\,\nabla_b R[\nabla]_{cae}{}^{f} - R[\nabla]_{cae}{}^{f}\,\nabla_b T_{df} - R[\nabla]_{cad}{}^{f}\,\nabla_b T_{fe} - R[\nabla]_{bae}{}^{f}\,\nabla_c T_{df} - R[\nabla]_{bad}{}^{f}\,\nabla_c T_{fe} + \nabla_c\nabla_b\nabla_a T_{de} - R[\nabla]_{bac}{}^{f}\,\nabla_f T_{de} - R[\nabla]_{cba}{}^{f}\,\nabla_f T_{de}$

The Second Bianchi Identity

Here we seek to reproduce the Second Bianchi identity. We redefine our covariant derivative

```
DefCovD[CD[-a], {";", "∇"}]
```

```
** DefCovD: Defining covariant derivative CD[-a].

** DefTensor: Defining vanishing torsion tensor TorsionCD[a, -b, -c].

** DefTensor: Defining symmetric Christoffel tensor ChristoffelCD[a, -b, -c].

** DefTensor: Defining Riemann tensor
 RiemannCD[-a, -b, -c, d]. Antisymmetric only in the first pair.

** DefTensor: Defining non-symmetric Ricci tensor RicciCD[-a, -b].

** DefCovD:  Contractions of Riemann automatically replaced by Ricci.
```

We begin by establishing a term

```
term1 = Antisymmetrize[CD[-e][ RiemannCD[-c, -d, -b, a] ], {-c, -d, -e} ]
```

$$\frac{1}{6}\left(\nabla_c R[\nabla]_{deb}{}^{a} - \nabla_c R[\nabla]_{edb}{}^{a} - \nabla_d R[\nabla]_{ceb}{}^{a} + \nabla_d R[\nabla]_{ecb}{}^{a} + \nabla_e R[\nabla]_{cdb}{}^{a} - \nabla_e R[\nabla]_{dcb}{}^{a}\right)$$

We sum these,

```
bianchi2 = 3 term1 // ToCanonical
```

$$\nabla_c R[\nabla]_{deb}{}^{a} - \nabla_d R[\nabla]_{ceb}{}^{a} + \nabla_e R[\nabla]_{cdb}{}^{a}$$

We expand the covariant derivatives in Christoffel symbols:

```
exp1 = bianchi2 // CovDToChristoffel
```

$$\Gamma[\nabla]^{a}{}_{ef}\ R[\nabla]_{cdb}{}^{f} - \Gamma[\nabla]^{f}{}_{eb}\ R[\nabla]_{cdf}{}^{a} - \Gamma[\nabla]^{a}{}_{df}\ R[\nabla]_{ceb}{}^{f} + \Gamma[\nabla]^{f}{}_{db}\ R[\nabla]_{cef}{}^{a} + \Gamma[\nabla]^{f}{}_{de}\ R[\nabla]_{cfb}{}^{a} - \Gamma[\nabla]^{f}{}_{ed}\ R[\nabla]_{cfb}{}^{a} + \Gamma[\nabla]^{a}{}_{cf}\ R[\nabla]_{deb}{}^{f} - \Gamma[\nabla]^{f}{}_{cb}\ R[\nabla]_{def}{}^{a} - \Gamma[\nabla]^{f}{}_{ce}\ R[\nabla]_{dfb}{}^{a} - \Gamma[\nabla]^{f}{}_{ec}\ R[\nabla]_{fdb}{}^{a} - \Gamma[\nabla]^{f}{}_{cd}\ R[\nabla]_{feb}{}^{a} + \Gamma[\nabla]^{f}{}_{dc}\ R[\nabla]_{feb}{}^{a} + \partial_{c}R[\nabla]_{deb}{}^{a} - \partial_{d}R[\nabla]_{ceb}{}^{a} + \partial_{e}R[\nabla]_{cdb}{}^{a}$$

We expand the Riemann tensors in Christoffel symbols

```
exp2 = exp1 // RiemannToChristoffel
```

$$
\begin{aligned}
&-\Gamma[\nabla]^{f}{}_{eb}\,\partial_c\Gamma[\nabla]^{a}{}_{df}+\Gamma[\nabla]^{f}{}_{db}\,\partial_c\Gamma[\nabla]^{a}{}_{ef}+\Gamma[\nabla]^{a}{}_{ef}\,\partial_c\Gamma[\nabla]^{f}{}_{db}-\\
&\Gamma[\nabla]^{a}{}_{df}\,\partial_c\Gamma[\nabla]^{f}{}_{eb}-\partial_c\partial_d\Gamma[\nabla]^{a}{}_{eb}+\partial_c\partial_e\Gamma[\nabla]^{a}{}_{db}+\Gamma[\nabla]^{f}{}_{eb}\,\partial_d\Gamma[\nabla]^{a}{}_{cf}-\\
&\Gamma[\nabla]^{f}{}_{eb}\left(\Gamma[\nabla]^{a}{}_{dg}\,\Gamma[\nabla]^{g}{}_{cf}-\Gamma[\nabla]^{a}{}_{cg}\,\Gamma[\nabla]^{g}{}_{df}-\partial_c\Gamma[\nabla]^{a}{}_{df}+\partial_d\Gamma[\nabla]^{a}{}_{cf}\right)-\\
&\Gamma[\nabla]^{f}{}_{cb}\,\partial_d\Gamma[\nabla]^{a}{}_{ef}-\Gamma[\nabla]^{a}{}_{ef}\,\partial_d\Gamma[\nabla]^{f}{}_{cb}+\\
&\Gamma[\nabla]^{a}{}_{ef}\left(\Gamma[\nabla]^{f}{}_{dg}\,\Gamma[\nabla]^{g}{}_{cb}-\Gamma[\nabla]^{f}{}_{cg}\,\Gamma[\nabla]^{g}{}_{db}-\partial_c\Gamma[\nabla]^{f}{}_{db}+\partial_d\Gamma[\nabla]^{f}{}_{cb}\right)+\\
&\Gamma[\nabla]^{a}{}_{cf}\,\partial_d\Gamma[\nabla]^{f}{}_{eb}+\partial_d\partial_c\Gamma[\nabla]^{a}{}_{eb}-\partial_d\partial_e\Gamma[\nabla]^{a}{}_{cb}-\Gamma[\nabla]^{f}{}_{db}\,\partial_e\Gamma[\nabla]^{a}{}_{cf}+\\
&\Gamma[\nabla]^{f}{}_{db}\left(\Gamma[\nabla]^{a}{}_{eg}\,\Gamma[\nabla]^{g}{}_{cf}-\Gamma[\nabla]^{a}{}_{cg}\,\Gamma[\nabla]^{g}{}_{ef}-\partial_c\Gamma[\nabla]^{a}{}_{ef}+\partial_e\Gamma[\nabla]^{a}{}_{cf}\right)+\\
&\Gamma[\nabla]^{f}{}_{cb}\,\partial_e\Gamma[\nabla]^{a}{}_{df}-\Gamma[\nabla]^{f}{}_{cb}\\
&\quad\left(\Gamma[\nabla]^{a}{}_{eg}\,\Gamma[\nabla]^{g}{}_{df}-\Gamma[\nabla]^{a}{}_{dg}\,\Gamma[\nabla]^{g}{}_{ef}-\partial_d\Gamma[\nabla]^{a}{}_{ef}+\partial_e\Gamma[\nabla]^{a}{}_{df}\right)+\Gamma[\nabla]^{a}{}_{df}\,\partial_e\Gamma[\nabla]^{f}{}_{cb}-\\
&\Gamma[\nabla]^{a}{}_{df}\left(\Gamma[\nabla]^{f}{}_{eg}\,\Gamma[\nabla]^{g}{}_{cb}-\Gamma[\nabla]^{f}{}_{cg}\,\Gamma[\nabla]^{g}{}_{eb}-\partial_c\Gamma[\nabla]^{f}{}_{eb}+\partial_e\Gamma[\nabla]^{f}{}_{cb}\right)-\\
&\Gamma[\nabla]^{a}{}_{cf}\,\partial_e\Gamma[\nabla]^{f}{}_{db}+\\
&\Gamma[\nabla]^{a}{}_{cf}\left(\Gamma[\nabla]^{f}{}_{eg}\,\Gamma[\nabla]^{g}{}_{db}-\Gamma[\nabla]^{f}{}_{dg}\,\Gamma[\nabla]^{g}{}_{eb}-\partial_d\Gamma[\nabla]^{f}{}_{eb}+\partial_e\Gamma[\nabla]^{f}{}_{db}\right)-\partial_e\partial_c\Gamma[\nabla]^{a}{}_{db}+\\
&\partial_e\partial_d\Gamma[\nabla]^{a}{}_{cb}+\Gamma[\nabla]^{f}{}_{de}\left(\Gamma[\nabla]^{a}{}_{fg}\,\Gamma[\nabla]^{g}{}_{cb}-\Gamma[\nabla]^{a}{}_{cg}\,\Gamma[\nabla]^{g}{}_{fb}-\partial_c\Gamma[\nabla]^{a}{}_{fb}+\partial_f\Gamma[\nabla]^{a}{}_{cb}\right)-\\
&\Gamma[\nabla]^{f}{}_{ed}\left(\Gamma[\nabla]^{a}{}_{fg}\,\Gamma[\nabla]^{g}{}_{cb}-\Gamma[\nabla]^{a}{}_{cg}\,\Gamma[\nabla]^{g}{}_{fb}-\partial_c\Gamma[\nabla]^{a}{}_{fb}+\partial_f\Gamma[\nabla]^{a}{}_{cb}\right)-\\
&\Gamma[\nabla]^{f}{}_{ec}\left(-\Gamma[\nabla]^{a}{}_{fg}\,\Gamma[\nabla]^{g}{}_{db}+\Gamma[\nabla]^{a}{}_{dg}\,\Gamma[\nabla]^{g}{}_{fb}+\partial_d\Gamma[\nabla]^{a}{}_{fb}-\partial_f\Gamma[\nabla]^{a}{}_{db}\right)-\\
&\Gamma[\nabla]^{f}{}_{ce}\left(\Gamma[\nabla]^{a}{}_{fg}\,\Gamma[\nabla]^{g}{}_{db}-\Gamma[\nabla]^{a}{}_{dg}\,\Gamma[\nabla]^{g}{}_{fb}-\partial_d\Gamma[\nabla]^{a}{}_{fb}+\partial_f\Gamma[\nabla]^{a}{}_{db}\right)-\\
&\Gamma[\nabla]^{f}{}_{cd}\left(-\Gamma[\nabla]^{a}{}_{fg}\,\Gamma[\nabla]^{g}{}_{eb}+\Gamma[\nabla]^{a}{}_{eg}\,\Gamma[\nabla]^{g}{}_{fb}+\partial_e\Gamma[\nabla]^{a}{}_{fb}-\partial_f\Gamma[\nabla]^{a}{}_{eb}\right)+\\
&\Gamma[\nabla]^{f}{}_{dc}\left(-\Gamma[\nabla]^{a}{}_{fg}\,\Gamma[\nabla]^{g}{}_{eb}+\Gamma[\nabla]^{a}{}_{eg}\,\Gamma[\nabla]^{g}{}_{fb}+\partial_e\Gamma[\nabla]^{a}{}_{fb}-\partial_f\Gamma[\nabla]^{a}{}_{eb}\right)
\end{aligned}
$$

We canonicalize this to enforce symmetries and antisymmetries,

```
exp3 = exp2 // ToCanonical
```

0

Which is what we expected.

$$\nabla_c R[\nabla]^{a}{}_{bde} + \nabla_d R[\nabla]^{a}{}_{bec} + \nabla_e R[\nabla]^{a}{}_{bcd} = 0$$

Perturbative GR

We begin this by opening the perturbative package:

```
<< xAct`xPert
```

Get::noopen : Cannot open xAct`xPert. >>

```
$Failed
```

We must first establish our metric perturbation, this is because the metric is a vacuum metric; thus all small perturbations are gravitational waves. We define the metric perturbation using the DefMetricPerturbation[the existing metric, name of the perturbation, perturbation parameter]

```
DefMetricPerturbation[metric, pert, ϵ]
```

```
** DefParameter: Defining parameter ϵ.
** DefTensor: Defining tensor pert[LI[order], -a, -b].
```

We can tell *Mathematica* to write the perturbation using the traditional notation *h*,

```
PrintAs[pert] ^= "h";
```

A second-order perturbation in the metric is then written

```
pert[LI[2], -a, -b]
```

$h^{2}{}_{ab}$

So the first-order perturbation of the metric g_{ab} is

```
Perturbation[metric[-a, -b], 1]
```

$h^{1}{}_{ab}$

for g^{ab} we have,

```
Perturbation[metric[a, b], 1]
```

$\triangle[g^{ab}]$

What? What is Δ? It turns out that *Mathematica* does not know how to perform a perturbation on the inverse metric. We need to use the ExpandPerturbation command,

```
Perturbation[metric[a, b], 1] // ExpandPerturbation
```

$-h^{1ab}$

This is not too impressive, let's try a fourth-order perturbation

```
Perturbation[metric[a, b], 4] // ExpandPerturbation
```

$$24\,h^{1ac}\,h^{1}{}_{c}{}^{d}\,h^{1}{}_{d}{}^{e}\,h^{1}{}_{e}{}^{b} - 12\,h^{1}{}_{c}{}^{d}\,h^{1}{}_{d}{}^{b}\,h^{2ac} + 6\,h^{2ac}\,h^{2}{}_{c}{}^{b} - 12\,h^{1ac}\,h^{1}{}_{d}{}^{b}\,h^{2}{}_{c}{}^{d} - 12\,h^{1ac}\,h^{1}{}_{c}{}^{d}\,h^{2}{}_{d}{}^{b} + 4\,h^{1}{}_{c}{}^{b}\,h^{3ac} + 4\,h^{1ac}\,h^{3}{}_{c}{}^{b} - h^{4ab}$$

Let's examine the third-order perturbed metric

```
Perturbed[metric[-a, -b], 3]
```

$$g_{ab} + \epsilon\,h^{1}{}_{ab} + \frac{1}{2}\epsilon^{2}\,h^{2}{}_{ab} + \frac{1}{6}\epsilon^{3}\,h^{3}{}_{ab}$$

and the inverse metric

```
Perturbed[metric[a, b], 3] // ExpandPerturbation
```

$$g^{ab} - \epsilon\, h^{1ab} + \frac{1}{2}\epsilon^2 \left(2\, h^{1ac}\ h^{1}{}_{c}{}^{b} - h^{2ab}\right) + \frac{1}{6}\epsilon^3 \left(-6\, h^{1ac}\ h^{1}{}_{c}{}^{d}\ h^{1}{}_{d}{}^{b} + 3\, h^{1}{}_{f}{}^{b}\ h^{2af} + 3\, h^{1ae}\ h^{2}{}_{e}{}^{b} - h^{3ab}\right)$$

Jolyon Bloomfield [5] produced a method of extracting only the first-order terms of the expansion, since ϵ is assumed to be small, where we can write the metric as the sum of a linearized metric and a perturbed metric, $g_{ab} = g^{o}{}_{ab} + h_{ab}$ we adapt it for our use

```
firstorderonly = pert[LI[n_], __] :> 0 /; n > 1;
Perturbed[metric[-a, -b], 3] /. firstorderonly
```

$$g_{ab} + \epsilon\, h^{1}{}_{ab}$$

Given an action we can derive an equation of motion. First we need to establish our scalar field,

```
DefTensor[sf[], M, PrintAs → "ϕ"]
** DefTensor: Defining tensor sf[].
```

We now define the perturbations of the scalar field,

```
DefTensorPerturbation[pertsf[LI[order]], sf[], M, PrintAs → "δϕ"]
```

```
** DefTensor: Defining tensor pertsf[LI[order]].
```

We now define the potential and the Planck mass,

```
DefScalarFunction[V]
DefConstantSymbol[massP, PrintAs → "mp"]
```

```
** DefScalarFunction: Defining scalar function V.
** DefConstantSymbol: Defining constant symbol massP.
```

Now we need to construct the Lagrangian

```
L = Sqrt[-Detmetric[]] (massP^2 / 2 RicciScalarcd[] - 1 / 2 cd[-b][sf[]] cd[b][sf[]] - V[sf[]])
```

$$\sqrt{-\tilde{\tilde{g}}}\left(\frac{m_p{}^2 R[\nabla]}{2} - V[\phi] - \frac{1}{2}\nabla_b\phi\,\nabla^b\phi\right)$$

We vary the fields

```
varL = L // Perturbation
```

$$-\left(\left(\triangle[\tilde{\tilde{g}}]\left(\frac{m_p^2 R[\nabla]}{2}-V[\phi]-\frac{1}{2}\nabla_b\phi\,\nabla^b\phi\right)\right)\Big/\left(2\sqrt{-\tilde{\tilde{g}}}\right)\right)+\sqrt{-\tilde{\tilde{g}}}\left(\frac{1}{2}m_p^2\,\triangle[R[\nabla]]+\frac{1}{2}\left(-\triangle\left[\nabla^b\phi\right]\nabla_b\phi-\delta\phi^1{}_{;b}\,\nabla^b\phi\right)-\delta\phi^1\,V'[\phi]\right)$$

We expand this

```
varL = L // Perturbation // ExpandPerturbation
```

$$-\left(\left(\tilde{\tilde{g}}\,h^1{}_a{}^a\left(\frac{m_p^2 R[\nabla]}{2}-V[\phi]-\frac{1}{2}\nabla_b\phi\,\nabla^b\phi\right)\right)\Big/\left(2\sqrt{-\tilde{\tilde{g}}}\right)\right)+\sqrt{-\tilde{\tilde{g}}}\left(\frac{1}{2}m_p^2\left(-h^{1ac}\,R[\nabla]_{ac}+g^{ac}\left(\frac{1}{2}\left(-h^{1d}{}_{d;c;a}-h^{1d}{}_{c;d;a}+h^1{}_{cd}{}^{;d}{}_{;a}\right)+\frac{1}{2}\left(h^{1d}{}_{c;a;d}+h^{1d}{}_{a;c;d}-h^1{}_{ca}{}^{;d}{}_{;d}\right)\right)\right)+\frac{1}{2}\left(-\delta\phi^1{}_{;b}\,\nabla^b\phi-\nabla_b\phi\left(g^{be}\,\delta\phi^1{}_{;e}-h^{1be}\,\nabla_e\phi\right)\right)-\delta\phi^1\,V'[\phi]\right)$$

We perform the metric contractions

```
varL = L // Perturbation // ExpandPerturbation // ContractMetric
```

$$-\frac{1}{2} m_p^2 \sqrt{-\tilde{\tilde{g}}}\; h^{1ab}\; R[\nabla]_{ab} + \frac{1}{4} m_p^2 \sqrt{-\tilde{\tilde{g}}}\; h^{1}{}_{a}{}^{a}\; R[\nabla] - \frac{1}{2} \sqrt{-\tilde{\tilde{g}}}\; h^{1}{}_{a}{}^{a}\; V[\phi] - \frac{1}{4} m_p^2 \sqrt{-\tilde{\tilde{g}}}\; h^{1b}{}_{b}{}^{;a}{}_{;a} - \frac{1}{4} m_p^2 \sqrt{-\tilde{\tilde{g}}}\; h^{1ba}{}_{;b;a} + \frac{1}{4} m_p^2 \sqrt{-\tilde{\tilde{g}}}\; h^{1a}{}_{b}{}^{;b}{}_{;a} + \frac{1}{2} \sqrt{-\tilde{\tilde{g}}}\; h^{1ba}\; \nabla_a\phi \nabla_b\phi + \frac{1}{2} m_p^2 \sqrt{-\tilde{\tilde{g}}}\; h^{1ba}{}_{;a;b} - \frac{1}{4} m_p^2 \sqrt{-\tilde{\tilde{g}}}\; h^{1a}{}_{a}{}^{;b}{}_{;b} - \frac{1}{2} \sqrt{-\tilde{\tilde{g}}}\; \nabla_b\phi\; \delta\phi^{1;b} - \frac{1}{2} \sqrt{-\tilde{\tilde{g}}}\; \delta\phi^{1}{}_{;b}\; \nabla^b\phi - \frac{1}{4} \sqrt{-\tilde{\tilde{g}}}\; h^{1}{}_{a}{}^{a}\; \nabla_b\phi \nabla^b\phi - \sqrt{-\tilde{\tilde{g}}}\; \delta\phi^1\; V'[\phi]$$

```
?RicciTo*
```

xAct`xTensor`

RicciToEinstein	RicciToTFRicci

RicciToEinstein[expr, covd] expands expr expressing all Ricci tensors of covd in terms of the Einstein and RicciScalar tensors of covd. If the second argument is a list of covariant derivatives the command is applied sequentially on expr. RicciToEinstein[expr] expands all Ricci tensors.

Then we enforce the symmetries

```
varL = L // Perturbation // ExpandPerturbation // ContractMetric // ToCanonical
```

$$-\frac{1}{2} m_p^2 \sqrt{-\tilde{\tilde{g}}}\; h^{1ba}\; R[\nabla]_{ba} + \frac{1}{4} m_p^2 \sqrt{-\tilde{\tilde{g}}}\; h^{1b}{}_{b}\; R[\nabla] - \frac{1}{2} \sqrt{-\tilde{\tilde{g}}}\; h^{1b}{}_{b}\; V[\phi] - \frac{1}{2} m_p^2 \sqrt{-\tilde{\tilde{g}}}\; h^{1b}{}_{b}{}^{;a}{}_{;a} + \frac{1}{2} m_p^2 \sqrt{-\tilde{\tilde{g}}}\; h^{1ba}{}_{;b;a} - \sqrt{-\tilde{\tilde{g}}}\; \nabla_b\phi\; \delta\phi^{1;b} + \frac{1}{2} \sqrt{-\tilde{\tilde{g}}}\; h^{1}{}_{ba}\; \nabla^a\phi \nabla^b\phi - \frac{1}{4} \sqrt{-\tilde{\tilde{g}}}\; h^{1a}{}_{a}\; \nabla_b\phi \nabla^b\phi - \sqrt{-\tilde{\tilde{g}}}\; \delta\phi^1\; V'[\phi]$$

The scalar equation of motion is then found by taking the variational derivatives of the scalar field and set them equal to 0

```
0 == VarD[pertsf[LI[1]], cd][varL] / Sqrt[-Detmetric[]]
```

$$0 == \frac{1}{\sqrt{-\tilde{\tilde{g}}}} \left(\delta_1{}^1 \sqrt{-\tilde{\tilde{g}}}\ \nabla_a \nabla^a \phi - \delta_1{}^1 \sqrt{-\tilde{\tilde{g}}}\ V'[\phi] \right)$$

we enforce the Kronecker delta

```
0 == VarD[pertsf[LI[1]], cd][varL] / Sqrt[-Detmetric[]] /. delta[-LI[1], LI[1]] → 1
```

$$0 == \frac{\sqrt{-\tilde{\tilde{g}}}\ \nabla_a \nabla^a \phi - \sqrt{-\tilde{\tilde{g}}}\ V'[\phi]}{\sqrt{-\tilde{\tilde{g}}}}$$

We canonicalize it

```
0 == VarD[pertsf[LI[1]], cd][varL] / Sqrt[-Detmetric[]] /.
  delta[-LI[1], LI[1]] → 1 // ToCanonical
```

$$0 == \nabla_a \nabla^a \phi - V'[\phi]$$

There we have it.

We can get the tensor equation of motion,

```
vartf = 2 (-VarD[pert[LI[1], a, b], cd][varL] / Sqrt[-Detmetric[]])
```

$$\frac{1}{\sqrt{-\tilde{\tilde{g}}}}\,2\left(\frac{1}{2}\,m_p^2\,\delta_1{}^1\,\sqrt{-\tilde{\tilde{g}}}\,R[\nabla]_{ab}-\frac{1}{4}\,m_p^2\,\delta_1{}^1\,\sqrt{-\tilde{\tilde{g}}}\,g_{ab}\,R[\nabla]+\frac{1}{2}\,\delta_1{}^1\,\sqrt{-\tilde{\tilde{g}}}\,g_{ab}\,V[\phi]+\frac{1}{4}\,\delta_1{}^1\,\sqrt{-\tilde{\tilde{g}}}\,g_{ab}\,\nabla_c\phi\,\nabla^c\phi-\frac{1}{2}\,\delta_1{}^1\,\sqrt{-\tilde{\tilde{g}}}\,g_{ac}\,g_{bd}\,\nabla^c\phi\,\nabla^d\phi\right)$$

This is a mess, so we begin enforcing the Kronecker deltas

```
vartf =
 2 (-VarD[pert[LI[1], a, b], cd][varL] / Sqrt[-Detmetric[]] /. delta[-LI[1], LI[1]] → 1)
```

$$\frac{1}{\sqrt{-\tilde{\tilde{g}}}}\,2\left(\frac{1}{2}\,m_p^2\sqrt{-\tilde{\tilde{g}}}\,R[\nabla]_{ab}-\frac{1}{4}\,m_p^2\sqrt{-\tilde{\tilde{g}}}\,g_{ab}\,R[\nabla]+\frac{1}{2}\sqrt{-\tilde{\tilde{g}}}\,g_{ab}\,V[\phi]+\frac{1}{4}\sqrt{-\tilde{\tilde{g}}}\,g_{ab}\,\nabla_c\phi\,\nabla^c\phi-\frac{1}{2}\sqrt{-\tilde{\tilde{g}}}\,g_{ac}\,g_{bd}\,\nabla^c\phi\,\nabla^d\phi\right)$$

We now expand by expressing all Ricci tensors of covariant derivatives in terms of Einstein tensors and Ricci scalars

```
vartf =
 2 (-VarD[pert[LI[1], a, b], cd][varL] / Sqrt[-Detmetric[]] /. delta[-LI[1], LI[1]] → 1 //
     SeparateMetric[metric] // RicciToEinstein)
```

$$\frac{1}{\sqrt{-\tilde{\tilde{g}}}} 2 \left(-\frac{1}{4} m_p{}^2 \sqrt{-\tilde{\tilde{g}}}\; g_{ab}\; R[\nabla] + \frac{1}{2} m_p{}^2 \sqrt{-\tilde{\tilde{g}}} \left(G[\nabla]_{ab} + \frac{1}{2} g_{ba}\; R[\nabla]\right) + \frac{1}{2} \sqrt{-\tilde{\tilde{g}}}\; g_{ab}\; V[\phi] - \frac{1}{2} \sqrt{-\tilde{\tilde{g}}}\; \nabla_a\phi\, \nabla_b\phi + \frac{1}{4} \sqrt{-\tilde{\tilde{g}}}\; g_{ab}\; g^{cd}\; \nabla_c\phi\, \nabla_d\phi\right)$$

We expand this

```
vartf =
 2 (-VarD[pert[LI[1], a, b], cd][varL] / Sqrt[-Detmetric[]] /. delta[-LI[1], LI[1]] → 1 //
      SeparateMetric[metric] // RicciToEinstein) // Expand
```

$$m_p{}^2\; G[\nabla]_{ab} - \frac{1}{2} m_p{}^2\; g_{ab}\; R[\nabla] + \frac{1}{2} m_p{}^2\; g_{ba}\; R[\nabla] + g_{ab}\; V[\phi] - \nabla_a\phi\, \nabla_b\phi + \frac{1}{2}\; g_{ab}\; g^{cd}\; \nabla_c\phi\, \nabla_d\phi$$

Then we enforce the symmetries,

```
vartf =
 2 (-VarD[pert[LI[1], a, b], cd][varL] / Sqrt[-Detmetric[]] /. delta[-LI[1], LI[1]] → 1 //
       SeparateMetric[metric] // RicciToEinstein) //
   Expand // ContractMetric // ToCanonical
```

$m_p{}^2\ G[\nabla]_{ab} + g_{ab}\ V[\phi] - \nabla_a\phi\,\nabla_b\phi + \frac{1}{2}\ g_{ab}\ \nabla_c\phi\,\nabla^c\phi$

We set this equal to 0,

```
0 == vartf
```

$0 == m_p{}^2\ G[\nabla]_{ab} + g_{ab}\ V[\phi] - \nabla_a\phi\,\nabla_b\phi + \frac{1}{2}\ g_{ab}\ \nabla_c\phi\,\nabla^c\phi$

Here we get the conservation of energy

```
eterm = IndexCollect[cd[a]@vartf // ToCanonical, CD[-b][sf[]]] // Simplify
```

$\nabla_b\phi\ (-\nabla_a\nabla^a\phi + V'[\phi])$

then

```
0 == eterm
```

$0 == \nabla_b\phi\,(-\nabla_a\nabla^a\phi + V'[\phi])$

this is degenerate with the scalar equation of motion

$0 == \nabla_a\nabla^a\phi - V'[\phi]$

Bibliography

[1] M. Nakahara, (2003), Geometry, Topology, and Physics, 2nd Edition, Taylor and Francis Group.
[2] Theodore Frankel, (2012), The Geometry of Physics, 3rd Edition, Cambridge University Press
[3] Jose M. Martin-Garcia, (2009), xAct Intro
[4] B. F. Schutz, (1984), The Use of Perturbation and Approximation Methods in General Relativity. In: Relativistic Astrophysics and Cosmology: Proceedings of the GIFT International Seminar on Theoretical Physics, (Eds.) Fustero, Xavier; Verdaguer, Enric; Singapore: World Scientific 35-98 (1984)
[5] Jolyon Bloomfield, (2013), xAct Tutorial

xAct can be found at the website: http://www.xact.es/index.html

Thank You for Attending!